

## What is Random Anyway?

by Michael Tempel, Director of Educational Services,  
Logo Computer Systems Inc., New York, New York

A number of people have run into an interesting problem while using **random** in the Apple versions of LogoWriter: What is random anyway? To illustrate this problem, and a solution to it, take a look at the following program for a target game. The game sets turtle 0 as a target. The player's goal is to get the turtle 1 to touch the target.

```
to game
  setup
  play
end

to setup
  rg ct setc 2 pu
  rt random 360 fd random 100
  tell 1 st pu
end

to play
  if colorunder = 2
    [print [YOU WIN] stop]
  fd 1
  if key? [direct readchar]
  play
end

to direct :key
  if :key = "l [lt 15]
  if :key = "r [rt 15]
end
```

The subprocedure **setup** places turtle 0 (the target) at a random spot on the screen. **Rt random 360** points it in a random direction and **fd random 100** moves it forward some amount less than 100 steps.

If you play the game again, **setup** will set turtle 0 to a new (random) starting position. But each time you start up LogoWriter and play the game, you will get the same random starting position. In other words, the first and second game in each LogoWriter session will always have the same starting position as the first and second game in any other session. Is this random?

While **random** does generate an unpredictable series of numbers, it always begins at the same point in this sequence each time you start LogoWriter. Try this: start LogoWriter, go to a new page and give the instruction:

```
repeat 10 [print random 50]
```

Write down the numbers you see. Start LogoWriter again and do the same thing. You'll get the same numbers. How can you get around this?

Write this procedure:

```
to randomize
  print random 100
  randomize
end
```

Before playing the game for the first time, run the procedure **randomize**. You'll see numbers rapidly printed on the screen. After a few moments, press the **stop** keys<sup>1</sup>. Now play your game. Do this each time you start up LogoWriter. It's very unlikely that you'll press the stop keys at exactly the same point in the sequence more than once. By using up some uncertain number of randoms, you make the starting position of your game unpredictable. However, you really don't want to see all those numbers printed on the page. In fact, you don't want to do anything with them, just use them up. Try this:

```
to randomize
  do.nothing.with random 100
  randomize
end
```

```
to do.nothing.with :something
end
```

**Do.nothing.with** takes an input, just like **print**, but it doesn't do anything with that input. However, since **random** is called upon to provide that input, the random number sequence moves along. When you press the **stop** keys, you are ready to play the game.

Incorporate the **randomize** procedure into the game.

```
to game
  randomize
  setup
  play
end
```

You'll need to add a line to stop **randomize** when a key is pressed, instead of using the stop keys.

<sup>1</sup> ⌘ S on the Apple, and CTRL S on the IBM.